
1 Introduction

Gretl supports two variants of the random effects estimator for panel data. These variants differ
in respect of the method for arriving at an estimate of the variance of the individual effects. The
default method in gretl is that of Swamy and Arora while the alternative is that of Nerlove (invoked
via the --nerlove option flag to the panel command).

We will write vi for the individual effects and σ2
v for their variance. As is well known, the method of

Swamy and Arora (1972) is to infer σ2
v from estimates of the Within and Between variances (obtained

from the residual variance of the fixed-effects model and the Between model, respectively). If we
write σ2

ε and σ2
b for the Within and Between variances respectively, the formula is

σ̂2
v = σ̂2

b − σ̂2
ε/T (1)

where T is the number of observations per individual, which at this point is assumed to be the same
for all individuals. The method of Nerlove (1971) is to infer σ2

v from the variance of the fixed effects
(that is, the estimated per-individual intercepts). Writing the fixed effects as αi, the formula is

σ̂2
v =

1

N − 1

N∑
i=1

(α̂i − ¯̂α)2 (2)

where N is the number of included individuals and ¯̂α is the mean of the intercepts.

Both of these methods were orginally developed with balanced samples (uniform T ) in mind. Baltagi
and Chang (1994) put foward a version of the Swamy–Arora method for unbalanced panels, further
discussion of which can be found in Baltagi (2005). To our knowledge, nobody has published a
corresponding unbalanced-panel version of Nerlove’s method, though one can envisage such a thing
by analogy with Baltagi and Chang’s modification of Swamy–Arora.

The purpose of this note is, first, to explicate the proposal of Baltagi and Chang (which is quite
easily misinterpreted) and to test two possible readings of their proposal against the “standard”
Swamy–Arora method in the context of unbalanced panels, and second, to construct and test a
similar modification to Nerlove’s method.

2 Understanding Baltagi–Chang

To establish notation, in an unbalanced panel we will refer to the number of usable observations for
individual i as Ti, and by n we will mean the total usable observations, n =

∑N
i=1 Ti.

The equation given by Baltagi and Chang for the unbalanced-panel version of Swamy–Arora is (using
our notation in the subscripts of the variance components) as follows:

σ̂2
v =

ûb′Pûb − (N −K)σ̂2
ε

n− tr((X ′PX)−1X ′ZZ ′X)
(3)

The matrix P is block-diagonal, the blocks being matrices of dimension Ti each of whose elements
are 1/Ti (so that multiplication by this matrix has the effect of obtaining Ti copies of the individual
mean of the multiplicand). N is as defined above, and K is the number of regressors including the
intercept.

The matrix whose trace figures in the denominator in (3) need not concern us here; its definition is
clear in context.1 The term that calls for clarification (in that it may be liable to misinterpretation) is
the leftmost element in the numerator, namely ûb′Pûb. Baltagi and Chang mean by ûb the residuals
from the Between regression, but it is important to note that in their presentation this regression
uses n observations, not just N (the number of included individuals) as is common in panel-data
software. On page 70 of their article they state that the Between estimator is obtained as

β̂b = (X ′PX)−1X ′Py (4)

Here X and y are of full length n, and the n × n matrix P is as described above.2 Since the
individual-mean values each appear Ti times in this regression, it is numerically identical to weighted
least squares, with weights Ti, on a dataset in which the individual-mean values appear just once.
This explains their comment (p. 73) that “ûb′Pûb can be obtained as the OLS residual sum of
squares from the regression involving

√
Tiȳi. on

√
TiX̄i.” If we were to think of the residuals ûb as

derived from a regression using N observations3 this comment would be wrong (or correct only in
the balanced case); but if the Between regression is defined as in (4) it is quite correct. In case a
demonstration of this point is required, it is given in Listing 1 in the Appendix.

The proprietary econometric software Stata is in the enviable position of being regarded as canonical
by many practitioners. Nonetheless, the interpretation of the Baltagi–Chang unbalanced variant of
Swamy–Arora embedded in Stata’s xtreg command (invoked with its sa option) is mistaken. What
Stata in fact calculates in place of ûb′Pûb is

SSR∗b =

N∑
i=1

Tiũ
2
i (5)

where ũi (i = 1, . . . , N) denotes the residual from a Between regression of length N . This is
therefore a Ti-weighted sum of squared residuals from an unweighted regression, as opposed to the
sum of squared residuals from a Ti-weighted regression, as per Baltagi–Chang: these quantities are
in general not equal (unless the Ti values are all the same, which ex hypothesi is not the case here).
That (5) truly represents what Stata does was determined by emulating the results from xtreg with
its sa option in gretl: if and only if we use this expression do we get exact numerical agreement.4

In a further twist to the tale, however, we cannot take for granted that what Baltagi and Chang
recommend is “right” (and what Stata does is “wrong”). That remains to be determined.

3 The simulation dataset

Our artificial dataset comprises a maximum of 20 observations on each of N = 10, 20 or 40 individ-
uals. In each case the individuals are divided into 5 classes of equal size, having Ti values of 1, 4,
10, 15 and 20 respectively. The mean Ti is therefore 10, so the total number of usable observations
is n = 10N (that is, 100, 200 or 400).

The data-generating process is
yit = α+ βxit + vi + εit

1But briefly, Z is an n × N matrix defined as diag(ıTi
), where ıTi

is a column vector of 1s of length Ti. When
multiplied into an N -vector u it produces an n-vector in which element i of u occurs Ti times.

2This is not specific to the 1994 Baltagi–Chang article; it is also the presentation given in Baltagi (2005) (and
other editions of Baltagi’s book).

3This would require that P be an N ×N diagonal matrix with the Ti’s on the diagonal.
4In addition this point is now clear in the documentation for version 15 of Stata (xtreg, “Methods and formulas”).

In previous versions the account given was inaccurate.

where xit ∼ N(0, 1), vi ∼ N(0, σ2
v) and εit ∼ N(0, σ2

ε). These three variates are both mutually and
serially independent.

For each run of the simulation the xit values are generated once while the vi and εit are generated
anew at each of K = 100,000 iterations. The runs differ in two dimensions: (a) the number of indi-
viduals, as mentioned above, and (b) the relative size of the individual and idiosyncratic variances,
namely

1. σ2
v = σ2

ε = 1.0;

2. σ2
v = 1.6, σ2

ε = 0.4; and

3. σ2
v = 0.4, σ2

ε = 1.6.

These choices conserve the variance of the composite error vi +εit at 2.0 while enabling us to explore
the effect of the relative size of its components. The Tables below therefore show 9 “cases”: three
sample sizes × three ratios of σ2

v to σ2
ε .

4 Simulation and results

In each run we employ four estimators, as follows.

1. What we will call the “standard” estimator (“std” for short). This employs a minimally
modified version of equation (1), in which T (undefined for an unbalanced panel) is replaced
by some sort of average of the Ti’s. This is used by Stata when the re option is given with
xtreg, and it is the default in gretl. Stata replaces T with the harmonic mean of the Ti’s. Since
the value in question appears as a divisor in the formula for σ̂2

v this choice seems reasonable
and gretl follows suit.5

2. Stata’s take on Baltagi–Chang, which, as noted above, uses in (3) the Ti-weighted sum of
squared residuals from an unweighted Between regression of length N . We label this “SBC.”

3. The estimator actually specified by Baltagi and Chang, which involves Ti-weighted estimation
of the Between model (labeled “BC”).

4. By way of benchmark, “Infeasible GLS” (IGLS): we can implement this since we know the
population values of σ2

v and σ2
ε and so can use these values in place of estimates.

We consider three figures of merit for the estimators: the mean error and MSE of σ̂v plus the MSE
of β̂, expressed as a ratio to that produced by IGLS. To be explicit, we mean by the MSE of β̂ the
mean value of (β̂ − β)2, or the mean square error of estimate, and similarly for the MSE of σ̂v.

Results are shown in Table 1.6 The best values for each “case” (minimum absolute value of mean
error, minima of the MSEs) are highlighted in red. These results enable us to address two questions:
(1) Is one of the Baltagi–Chang variants clearly “correct” and the other ill advised? (2) Does either
of these variants offer a demonstrable efficiency gain over the standard method?

On the first question—comparison of BC and SBC—it is clear that BC is a more efficient estimator of
the variance of the individual effects than SBC. The figures of merit for σ̂2

v are always substantially
better for BC than SBC (although MSE(σ̂2

v) for BC is in some cases inferior to that for the std
method). However, this does not translate into greater efficiency in estimation of the slope coefficient,
β, which is presumably of greater inherent interest in practice: in fact, SBC is slightly better in that
regard. In all 9 cases in Table 1 SBC gives a lesser MSE(β̂) than BC, although the differences are
quite small (and SBC is in two cases marginally inferior to std in this respect).

Cameron and Trivedi (2005) remark of the random effects model that “more efficient estimators

of the variance components. . . will not necessarily increase the efficiency of β̂RE” (p. 734). We’re

5We have not been able to find a source in the panel-data literature for this recommendation.
6The script that produced this output is shown in Listing 2 in the Appendix.

ME(σ̂2
v) MSE(σ̂2

v) MSE(β̂)/IGLS

N = 10, n = 100

σ2
v = σ2

ε std −0.0857 0.3800 1.00908

SBC +0.1139 0.4601 1.00405

BC −0.0021 0.3870 1.00483

IGLS 0.0000 0.0000 1.00000

σ2
v > σ2

ε std −0.0370 0.7081 1.00596

SBC +0.1508 1.0320 1.00475

BC −0.0029 0.8888 1.00595

IGLS 0.0000 0.0000 1.00000

σ2
v < σ2

ε std −0.0960 0.1486 1.01495

SBC +0.0782 0.1193 1.00345

BC +0.0004 0.0905 1.00404

IGLS 0.0000 0.0000 1.00000

N = 20, n = 200

σ2
v = σ2

ε std −0.0337 0.1872 1.00313

SBC +0.0783 0.2063 1.00200

BC +0.0010 0.1783 1.00228

IGLS 0.0000 0.0000 1.00000

σ2
v > σ2

ε std −0.0134 0.3239 1.00135

SBC +0.1070 0.4670 1.00136

BC +0.0014 0.4097 1.00155

IGLS 0.0000 0.0000 1.00000

σ2
v < σ2

ε std −0.0448 0.0959 1.01982

SBC +0.0495 0.0519 1.00307

BC +0.0006 0.0424 1.00391

IGLS 0.0000 0.0000 1.00000

N = 40, n = 400

σ2
v = σ2

ε std −0.0140 0.0916 1.00146

SBC +0.0401 0.0933 1.00124

BC +0.0001 0.0871 1.00139

IGLS 0.0000 0.0000 1.00000

σ2
v > σ2

ε std −0.0062 0.1545 1.00057

SBC +0.0532 0.2127 1.00073

BC +0.0001 0.2007 1.00079

IGLS 0.0000 0.0000 1.00000

σ2
v < σ2

ε std −0.0203 0.0528 1.01346

SBC +0.0271 0.0229 1.00229

BC +0.0001 0.0205 1.00269

IGLS 0.0000 0.0000 1.00000

Table 1: Errors of estimation, Swamy–Arora (ME = mean error, MSE = mean square error)

seeing a noteworthy instance of their point here: a clearly inefficient estimator of σ2
v , SBC, turns

out to yield the most efficient estimator of β. In each of the runs reported in Table 1 we initialized
gretl’s random number generator with the same seed for the sake of consistency. However, in light
of the somewhat surprising results we performed additional runs with different seeds (and also with
different patterns of imbalance in the data). On that basis we’re able to say that the reported results
are not flukey but quite typical.

Our second question concerned the gain from using either BC or SBC relative to the minimally
adjusted version of Swamy–Arora. Our results suggest that one can gain efficiency in estimation
of σ2

v by using BC (but not SBC). Both SBC and (to a lesser extent) BC appear to offer greater
efficiency in estimation of β, though the gain is quite modest and not guaranteed. In particular,
when σ2

v is large relative to σ2
ε the “standard” estimator may work best; conversely, the greatest

gains from SBC and BC seem to occur when σ2
v is relatively small.

5 Nerlove’s method

The modification to Swamy–Arora proposed by Baltagi and Chang involved weighting by the Ti
values in an unbalanced panel. In the case of Nerlove’s estimator, a “natural” analogy would be to
compute the Ti-weighted variance of the estimated fixed effects. That is, we would replace equation
(2) with

σ̂2
v =

N

N − 1

N∑
i=1

wi(α̂i − ¯̂αw)2

where

wi = Ti/n and ¯̂αw =

N∑
i=1

wiα̂i

Table 2 shows results from simulations on the same pattern as those performed in relation to Swamy–
Arora. Here “std” means the standard unweighted formula and “wtd” the Ti-weighted one. Our
findings are broadly similar to those for Swamy–Arora: efficiency may be gained in estimation
of both σ2

v and β by weighting, but to the greatest extent when σ2
v is relatively small, while the

unweighted estimator may be superior when σ2
v is large.

6 Conclusion

We would like to be able to draw definite conclusions regarding the best policy for gretl to adopt,
but this turns out to be difficult. We should probably offer as an option an unbalanced-sample
variant of Swamy–Arora beyond the minimal “standard” version, but should it be that of Baltagi
and Chang, or that popularized by Stata?

Secondly, are we confident enough about our own unbalanced-sample variant of the Nerlove estimator
to offer it as an option? It can reduce efficiency when σ2

v is large in relation to σ2
ε so maybe we

would not be doing users a favor?

ME(σ̂2
v) MSE(σ̂2

v) MSE(β̂)/IGLS

N = 10, n = 100

σ2
v = σ2

ε std +0.2888 0.4816 1.00345

wtd +0.0499 0.3400 1.00205

IGLS 0.0000 0.0000 1.00000

σ2
v > σ2

ε std +0.1141 0.6731 1.00134

wtd −0.0408 0.7771 1.00302

IGLS 0.0000 0.0000 1.00000

σ2
v < σ2

ε std +0.4642 0.4527 1.00837

wtd +0.1409 0.0997 1.00229

IGLS 0.0000 0.0000 1.00000

N = 20, n = 200

σ2
v = σ2

ε std +0.2878 0.2714 1.00170

wtd +0.0777 0.1733 1.00122

IGLS 0.0000 0.0000 1.00000

σ2
v > σ2

ε std +0.1158 0.3259 1.00058

wtd +0.0036 0.3858 1.00088

IGLS 0.0000 0.0000 1.00000

σ2
v < σ2

ε std +0.4594 0.3234 1.00513

wtd +0.1517 0.0623 1.00199

IGLS 0.0000 0.0000 1.00000

N = 40, n = 400

σ2
v = σ2

ε std +0.2848 0.1723 1.00175

wtd +0.0891 0.0915 1.00109

IGLS 0.0000 0.0000 1.00000

σ2
v > σ2

ε std +0.1135 0.1645 1.00050

wtd +0.0220 0.1935 1.00059

IGLS 0.0000 0.0000 1.00000

σ2
v < σ2

ε std +0.4565 0.2629 1.00830

wtd +0.1561 0.0439 1.00315

IGLS 0.0000 0.0000 1.00000

Table 2: Errors of estimation, Nerlove (ME = mean error, MSE = mean square error)

Appendix: hansl scripts

Listing 1: Demonstration of equivalence

This script demonstrates the equivalence of the two ways given by Baltagi and Chang to obtain
the term they notate as ûb′Pûb: (a) quite literally, where ûb is the residual from a full-length
Between regression and P is a block-diagonal matrix as described in the text, and (b) as the sum of
squared residuals from a Between regression of length N , where all the (individual-mean) data are
multiplied by

√
Ti. The script also demonstrates that this magnitude differs from the Ti-weighted

sum of squared residuals from an unweighted Between regression of length N , as employed by Stata.

set verbose off

nulldata 120

# panel structure: max(Ti) = 4, 30 individuals

setobs 4 1:1 --stacked-time-series

series x = normal()

series vi = pexpand(mnormal(30,1))

series eit = normal()

series y = 10 + x + vi + eit

# apply mask to reduce Ti to 2 for the first 15 individuals

scalar j=1

loop i=1..15 -q

y[j] = NA

x[j] = NA

y[j+1] = NA

x[j+1] = NA

j += 4

endloop

# run "standard" between regression with N=30 observations

panel y 0 x --between

matrix ub = $uhat

ub[1:15] *= sqrt(2)

ub[16:30] *= sqrt(4)

printf "Ti-weighted SSR (a la Stata): %g\n", ub’ub

/*

# compare stata (same results as from gretl: uncomment if you have stata)

genr unit

genr time

foreign language=stata --send-data

xtset unit time

xtreg y x, be

end foreign

*/

Continued on next page.

Listing 1: Demonstration of equivalence, continued

# build the Baltagi-Chang P matrix (** = Kronecker product)

matrix P1 = I(15) ** ones(2,2)/2

matrix P2 = I(15) ** ones(4,4)/4

matrix P = (P1 ~ zeros(30,60)) | (zeros(60,30) ~ P2)

# calculate ub’P*ub as per Baltagi-Chang

matrix my = {y}

matrix X = ones(90,1) ~ {x}

matrix bb1 = inv(X’P*X) * X’P*my

printf "\nBaltagi-Chang between beta-hat (regression length n)"

printf "\n%#.6g", bb1

matrix ub = my - X*bb1

printf "ub’P*ub = %g\n", ub’P*ub

# construct individual-means dataset of length N, and run

# Ti-weighted between regression

matrix ybar = pshrink(pmean(y))

matrix Xbar = ones(30,1) ~ pshrink(pmean(x))

ybar[1:15] *= sqrt(2)

ybar[16:30] *= sqrt(4)

Xbar[1:15,] *= sqrt(2)

Xbar[16:30,] *= sqrt(4)

matrix bb2 = mols(ybar, Xbar)

printf "\nbeta-hat from weighted between regression (length N)"

printf "\n%#.6g", bb2

matrix ubw = ybar - Xbar*bb2

printf "ubw’ubw = %g\n", ubw’ubw

Listing 2: Simulation script

This is the script used to populate Table 1 in the text. It is designed to be called by the program
gretlcli with the --scriptopt command-line option set to 1, 2, or 3 (for N = 10), 21, 22 or 23
(N = 20), or 41, 42, or 43 (N = 40). The Nerlove table is produced by a very similar script.

set verbose off

scalar parms = scriptopt

scalar N = 10

if parms > 40

N = 40

parms -= 40

elif parms > 20

N = 20

parms -= 20

endif

scalar T = 20 # max time-series length

scalar NT = N*T # total length of dataset

nulldata NT --preserve

setobs T 1:1 --stacked-time-series

set seed 555312 # comment this out for a runtime-based seed

matrix Ti = {1, 5, 10, 14, 20}

series x = normal()

scalar K = 100000

# set the relative variances based on @parms

if parms == 1

true_s2v = 1.0

true_s2e = 1.0

elif parms == 2

true_s2v = 1.6

true_s2e = 0.4

else

true_s2v = 0.4

true_s2e = 1.6

endif

scalar true_sv = sqrt(true_s2v)

scalar true_se = sqrt(true_s2e)

scalar true_b2 = 1.0

printf "\ntrue_s2v = %.1f\n", true_s2v

printf "true_s2e = %.1f\n", true_s2e

printf "true_b2 = %.1f\n", true_b2

printf "NT = %d\n", NT

printf "K = %d\n", K

matrix s2v = zeros(K,4)

matrix b2 = zeros(K,4)

Continued on next page.

Listing 2: Simulation script, continued

# the number of distinct Ti values

scalar nTi = nelem(Ti)

# the number of individuals with each Ti value

scalar indiv_per = N/nTi

# write matrix of population variances for IGLS

string vname = sprintf("reV%d.mat", parms)

mwrite({true_s2v, true_s2e}, vname)

loop k=1..K -q

series v = pexpand(mnormal(N,1) * true_sv)

series e = normal() * true_se

series y = 10 + true_b2*x + v + e

# unbalance the data!

loop j=1..nTi -q

scalar tmax = T - Ti[j]

loop i=1..indiv_per -q

scalar ij = 1 + (j-1)*T*indiv_per + (i-1) * T

# printf "j=%d, i=%d, ij=%d, tmax=%d\n", j, i, ij, tmax

loop t=1..tmax -q # time

y[ij+t-1] = NA

endloop

endloop

endloop

panel y 0 x --random --quiet

s2v[k,1] = $model.s2v

b2[k,1] = $coeff[2]

panel y 0 x --random --exact --quiet

s2v[k,2] = $model.s2v

b2[k,2] = $coeff[2]

panel y 0 x --random --exact=2 --quiet

s2v[k,3] = $model.s2v

b2[k,3] = $coeff[2]

panel y 0 x --random --exact=@vname --quiet

s2v[k,4] = $model.s2v

b2[k,4] = $coeff[2]

endloop

matrix ME = meanc(s2v - true_s2v)

matrix MSE = meanc((s2v - true_s2v).^2)

matrix bMSE = meanc((b2 - true_b2).^2)

printf "\nMEs, MSE of s2v estimate, b2 MSE ratio\n\n"

printf " & & ME & MSE & ratio\n"

printf " & std & %+.4f & %.4f & %.5f \\\\\n", ME[1], MSE[1], bMSE[1]/bMSE[4]

printf " & SBC & %+.4f & %.4f & %.5f \\\\\n", ME[2], MSE[2], bMSE[2]/bMSE[4]

printf " & BC & %+.4f & %.4f & %.5f \\\\\n", ME[3], MSE[3], bMSE[3]/bMSE[4]

printf " & IGLS & %4f & %.4f & %.5f \\\\\n", ME[4], MSE[4], bMSE[4]/bMSE[4]

11


	1 Introduction
	2 Understanding Baltagi–Chang
	3 The simulation dataset
	4 Simulation and results
	5 Nerlove's method
	6 Conclusion

